Добавил в типовую конфигурацию (УТ11) несколько самодельных отчетов СКД (не внешних). Их невозможно добавить в избранное:
- в интерфейсе "такси" отсутствует звездочка рядом с названием;
- в обычно интерфейсе отсутствует команда "добавить в избранное".
Видимо не включил где-то какую-то галку. Поиск по форуму результатов не дал. На что обратить внимание?
---
roman77, добавила ваш отчет к себе в конфигурацию (правда, в бухгалтерию, УТ нет), сделала обновление вспомогательных данных. Возможность добавить в избранное появилась.
---
roman77, Может формы нет?
---
dj_serega, формы нет, но у типовых отчетов тоже нет, а они в избранное помещаются.
---
roman77, а в подсистему отчет выведен?
---
roman77, несколько самодельных отчетов СКД (не внешних)
Как делали? Возможно что-то с ГУИД.
---
roman77, хм...интересный ситуэйшн...а можно отчет глянуть
---
roman77, проверил в БП есть така возможность. в УТ не могу проверить
---
roman77, платформа какая кстати...у меня 1С:Предприятие 8.3 (8.3.5.1383) на других и не смотрел.
Кстати, уточню - добавили в "дополнительные отчёты и обработки"(не знаю как это в УТ) выбрали разместить в разделе, выбрали раздел командного интерфейса, обозначили пользователей быстрого доступа ... открыли свой отчет и нет галки избранного?
---
ben19791010, нет. Это не внешние отчеты, а отчеты в самой конфигурации. Но если оформить отчет как внешний и подключить через "Дополнительные отчеты и обработки", то команды добавить в избранное все равно нету. Прикрепил пример такого отчета.
---
roman77, ТИИ делали?
---
roman77, добавила ваш отчет к себе в конфигурацию (правда, в бухгалтерию, УТ нет), сделала обновление вспомогательных данных. Возможность добавить в избранное появилась.
---
sonne, у меня не в этом было дело, но все равно спасибо за потраченное время.
---
roman77, считается правилом хорошего тона рассказать о самостоятельно найденной проблеме/решении.
---
PetroP, это был мой личный косяк.
---
roman77, и другой человек его не повторит? :)
---
roman77, в отчете в "Командах" поставьте галочку "Использовать стандартные команды".
---
roman77, Добавил в типовую конфигурацию (УТ11) несколько самодельных отчетов СКД (не внешних).
А почему не воспользовались стандартным механизмом БСП? Добавление команды и настройка видимости вполне себе удобно настраивается.
ИМХО, в конфигурацию стоит встраивать только те отчеты, которые заменяют типовые.
---
Добавил вознаграждение
---
А пробовали запустить обновление вспомогательных данных или что-то подобное?
---
- платформа 8.3.5.1460
- отчеты выведены в соответствующие подсистемы
- Галка "использовать стандартные команды" включена

После того, как я добавил отчеты в конфигурацию, я прогнал ОбновлениеВспомогательныхДанных.epf с опцией "идентификаторы объектов метаданных". Но без этого новые объекты в принципе не открываются.
---
Например, если есть форма и свойство "АвтоНавигационнаяСсылка" = Ложь, то избранное недоступно для данной формы. Но без формы по умолчанию должно быть...
---
Содержимым избранного можно управлять программно с помощью объекта встроенного языка ИзбранноеРаботыПользователя
Особенностью является то, что доступ к этому объекту не предоставляется через свойство глобального контекста, как, например, к истории работы пользователя.
Для того чтобы получить доступ к избранному, нужно прочитать из системного хранилища настроек избранное, добавить в него новый элемент с нужной ссылкой, и сохранить избранное обратно в системное хранилище настроек.
Следующий пример показывает добавление в избранное двух ссылок: ссылки на элемент справочника и внешней гиперссылки.
---
Надо в свойствах отчета выбрать хранилище вариантов отчетов
---
Спасибо. это помогло
---
Случилась аналогичная задача. 1С Документооборот 2.1. Внешний отчет открывался без "звездочки" избранного. Помогло определение формы отчета и формы варианта отчета (выбрала из типовых) - "звездочка" появилась.
---
Если отчета определена Форма отчета то выбрать хранилище вариантов отчета не достаточно.
---
А что еще нужно? Выбрал варианты, обновил метаданные. Звездочка так и не ставится. Что делать?
---
у самого задача висит. не знаю как реализовать
---
У кого во внешних дополнительных отчетах звёздочка в форме есть, но не ставится, это косяк в конфигурациях на БСП 2ххх редакции. В модуле формы отчета конфигурации, скорее всего, не хватает определения навигационной ссылки, ведущей на справочник Варианты отчетов, добавьте в "при создании на сервере":

```1c
    АвтоНавигационнаяСсылка = Ложь;
    НавигационнаяСсылка = ";
    Если ЗначениеЗаполнено(НастройкиОтчета.ВариантСсылка)
    И Не НастройкиОтчета.Внешний Тогда

    НавигационнаяСсылка = ПолучитьНавигационнуюСсылку(НастройкиОтчета.ВариантСсылка);

    КонецЕсли;
```

где НастройкиОтчета.ВариантСсылка хранит ссылку на "Варианты отчетов". На этом звёздочки уже будут ставиться.
Далее чтобы избранное открывало отчет, а не справочник вариантов, в самом справочнике вариантов в модуле менеджера надо переопределить обработчик получения формы

```1c
    Процедура ОбработкаПолученияФормы(ВидФормы, Параметры, ВыбраннаяФорма, ДополнительнаяИнформация, СтандартнаяОбработка)

    // Переопределение для целей избранного - вместо карточки с настройками размещения отчета будет открываться его
    // основная форма.
    Если ВидФормы = "ФормаОбъекта" Тогда
    СсылкаВарианта = ОбщегоНазначенияКлиентСервер.СвойствоСтруктуры(Параметры, "Ключ");
    Если Не ЗначениеЗаполнено(СсылкаВарианта) Тогда
    ВызватьИсключение НСтр("ru = 'Новый вариант отчета можно создать только из формы отчета'");
    КонецЕсли;
    ПоказатьКарточку = ОбщегоНазначенияКлиентСервер.СвойствоСтруктуры(Параметры, "ПоказатьКарточку");
    Если ПоказатьКарточку = Истина Тогда
    Возврат;
    КонецЕсли;

    ПараметрыОткрытия = ВариантыОтчетов.ПараметрыОткрытия(СсылкаВарианта);

    ВариантыОтчетовКлиентСервер.ДополнитьСтруктуруКлючом(ПараметрыОткрытия, "ВыполнятьЗамеры", Ложь);

    Если ПараметрыОткрытия.ТипОтчета = "Внутренний" Или ПараметрыОткрытия.ТипОтчета = "Расширение" Тогда
    Вид = "Отчет";
    ИначеЕсли ПараметрыОткрытия.ТипОтчета = "Дополнительный" Тогда
    Вид = "ВнешнийОтчет";
    Если Не ПараметрыОткрытия.Свойство("Подключен") Тогда
    ВариантыОтчетов.ПриПодключенииОтчета(ПараметрыОткрытия);
    КонецЕсли;
    Если Не ПараметрыОткрытия.Подключен Тогда
    ВызватьИсключение НСтр("ru = 'Вариант внешнего отчета можно открыть только из формы отчета.'");
    КонецЕсли;
    Иначе
    ВызватьИсключение НСтр("ru = 'Вариант внешнего отчета можно открыть только из формы отчета.'");
    КонецЕсли;

    ПолноеИмяОтчета = Вид + "." + ПараметрыОткрытия.ИмяОтчета;

    КлючУникальности = ПолноеИмяОтчета;
    Если ЗначениеЗаполнено(ПараметрыОткрытия.КлючВарианта) Тогда
    КлючУникальности = КлючУникальности + "/КлючВарианта." + ПараметрыОткрытия.КлючВарианта;
    КонецЕсли;

    ПараметрыОткрытия.Вставить("КлючПараметровПечати", КлючУникальности);
    ПараметрыОткрытия.Вставить("КлючСохраненияПоложенияОкна", КлючУникальности);

    СтандартнаяОбработка = Ложь;
    Если ПараметрыОткрытия.ТипОтчета = "Дополнительный" Тогда // Для платформы.
    ВыбраннаяФорма = "Справочник.ВариантыОтчетов.ФормаОбъекта";
    Параметры.Вставить("ПараметрыОткрытияФормыОтчета", ПараметрыОткрытия);
    Возврат;
    КонецЕсли;
    ВыбраннаяФорма = ПолноеИмяОтчета + ".Форма";
    ОбщегоНазначенияКлиентСервер.ДополнитьСтруктуру(Параметры, ПараметрыОткрытия);
    КонецЕсли;

    КонецПроцедуры
```

обработчик из новой Розницы, успешно запустился в древней УТ 11.2 без бубна.

а в модуле формы элемента справочника надо добавить:
1. при создании на сервере

```1c
    Если Параметры.Свойство("ПараметрыОткрытияФормыОтчета", ПараметрыОткрытияФормыОтчета) Тогда
    Возврат;
    КонецЕсли;
```

2. при открытии

```1c
    Если ПараметрыОткрытияФормыОтчета <> Неопределено Тогда
    Отказ = Истина;
    ВариантыОтчетовКлиент.ОткрытьФормуОтчета(Неопределено, ПараметрыОткрытияФормыОтчета);
    КонецЕсли;
```

3. создать в форме реквизит ПараметрыОткрытияФормыОтчета типа Произвольный
---
Причина может быть еще и в том, что у реквизита "Объект" формы не стоит флажок на свойстве "Основной реквизит". Если его поставить, то звездочка Избранного появится.
---
Ваш комент подтолкнул меня к размышлениям.
У меня БСП 2.3.6. Особенность что у отчета своя форма, а не общая. По сути я из общей перетащил самое необходимое:

Добавил реквизит формы НастройкиОтчета тип Произвольный

ПриСоздании добавил.

```1c
    // Локальные переменные.
    ОтчетОбъект = РеквизитФормыВЗначение("Отчет");
    ОтчетМетаданные = ОтчетОбъект.Метаданные();
    НастройкиОтчета = ОтчетыКлиентСервер.ПолучитьНастройкиОтчетаПоУмолчанию();
    АдресСхемы = ПоместитьВоВременноеХранилище(ОтчетОбъект.СхемаКомпоновкиДанных, УникальныйИдентификатор);

    НастройкиОтчета = ОтчетыКлиентСервер.ПолучитьНастройкиОтчетаПоУмолчанию();
    НастройкиОтчета.Вставить("КлючСхемы", ");
    НастройкиОтчета.Вставить("ПолноеИмя", ОтчетМетаданные.ПолноеИмя());
    НастройкиОтчета.Вставить("ОтчетСсылка", Неопределено);

    ТипОтчетаСтрокой = ВариантыОтчетовКлиентСервер.ТипОтчета(НастройкиОтчета.ОтчетСсылка, Истина);
    Если ТипОтчетаСтрокой = Неопределено Тогда
    Информация = ВариантыОтчетов.СформироватьИнформациюОбОтчетеПоПолномуИмени(НастройкиОтчета.ПолноеИмя);
    НастройкиОтчета.ОтчетСсылка = Информация.Отчет;
    КонецЕсли;
```

в ПриЗагрузкеВариантаНаСервере добавил

```1c
    ОбновитьИнформациюОВариантеОтчета();

    АвтоНавигационнаяСсылка = Ложь;
    НавигационнаяСсылка = ";
    Если ЗначениеЗаполнено(НастройкиОтчета.ВариантСсылка)
    Тогда
    НавигационнаяСсылка = ПолучитьНавигационнуюСсылку(НастройкиОтчета.ВариантСсылка);
    КонецЕсли;
```

и скопировал процедуру

```1c
    &НаСервере
    Процедура ОбновитьИнформациюОВариантеОтчета()

    Запрос = Новый Запрос;
    Запрос.Текст =
    "ВЫБРАТЬ РАЗРЕШЕННЫЕ ПЕРВЫЕ 1
    | ВариантыОтчетов.Ссылка КАК ВариантСсылка,
    | ВариантыОтчетов.ПредопределенныйВариант.КлючЗамеров КАК КлючЗамеров,
    | ВариантыОтчетов.ПредопределенныйВариант КАК ПредопределенныйСсылка,
    | ВЫБОР
    | КОГДА ВариантыОтчетов.Пользовательский
    | ИЛИ ВариантыОтчетов.Родитель.КлючВарианта ЕСТЬ NULL
    | ТОГДА ВариантыОтчетов.КлючВарианта
    | ИНАЧЕ ВариантыОтчетов.Родитель.КлючВарианта
    | КОНЕЦ КАК ИмяИсходногоВарианта,
    | ВариантыОтчетов.Пользовательский КАК Пользовательский
    |ИЗ
    | Справочник.ВариантыОтчетов КАК ВариантыОтчетов
    |ГДЕ
    | ВариантыОтчетов.Отчет = &Отчет
    | И ВариантыОтчетов.КлючВарианта = &КлючВарианта";
    Запрос.УстановитьПараметр("Отчет", НастройкиОтчета.ОтчетСсылка);
    Запрос.УстановитьПараметр("КлючВарианта", КлючТекущегоВарианта);

    Выборка = Запрос.Выполнить().Выбрать();

    Если Выборка.Следующий() Тогда
    НастройкиОтчета.Вставить("ВариантСсылка", Выборка.ВариантСсылка);
    НастройкиОтчета.Вставить("КлючЗамеров", Выборка.КлючЗамеров);
    НастройкиОтчета.Вставить("ПредопределенныйСсылка", Выборка.ПредопределенныйСсылка);
    НастройкиОтчета.Вставить("ИмяИсходногоВарианта", ?(Выборка.Пользовательский, Выборка.ИмяИсходногоВарианта, КлючТекущегоВарианта));
    НастройкиОтчета.Вставить("Пользовательский", Выборка.Пользовательский);
    Иначе
    НастройкиОтчета.Вставить("ВариантСсылка", Неопределено);
    НастройкиОтчета.Вставить("КлючЗамеров", Неопределено);
    НастройкиОтчета.Вставить("ПредопределенныйСсылка", Неопределено);
    НастройкиОтчета.Вставить("ИмяИсходногоВарианта", Неопределено);
    НастройкиОтчета.Вставить("Пользовательский", Неопределено);
    КонецЕсли;
    КонецПроцедуры
```
---
сделал проще:
1. в обработку добавил реквизит "НавигационнаяСсылка".
2. при создании на сервере добавил ссылку на эту обработку в справочнике ДополнительныеОтчетыИОбработки

```1c
    &НаСервере
    Процедура ПриСозданииНаСервере(Отказ, СтандартнаяОбработка)
    Объект.НавигационнаяСсылка = ПолучитьНавигационнуюСсылку(Справочники.ДополнительныеОтчетыИОбработки.НайтиПоРеквизиту("ИмяОбъекта","ИмяМоейОбработки"));
    КонецПроцедуры
```
---
немного поторопился :(
Звездочка "Избранное" появилась, но ведет на элемент справочника, а не на открытие внешней обработки...
---
Спасибо user1800274
Для унф 3.0. Добавление в избранное отчетов без СКД, со своей формой.
Код устанавливаем в модуль формы отчета.

```1c
    &НаСервере
    Процедура ПриСозданииНаСервере(Отказ, СтандартнаяОбработка)
    ОтчетОбъект = РеквизитФормыВЗначение("Объект");
    ОтчетМетаданные = ОтчетОбъект.Метаданные();
    НастройкиОтчета = ОтчетыКлиентСервер.ПолучитьНастройкиОтчетаПоУмолчанию();
    АдресСхемы = ПоместитьВоВременноеХранилище(ОтчетОбъект.СхемаКомпоновкиДанных, УникальныйИдентификатор);

    НастройкиОтчета = ОтчетыКлиентСервер.ПолучитьНастройкиОтчетаПоУмолчанию();
    НастройкиОтчета.Вставить("КлючСхемы", ");
    НастройкиОтчета.Вставить("ПолноеИмя", ОтчетМетаданные.ПолноеИмя());
    НастройкиОтчета.Вставить("ОтчетСсылка", Неопределено);

    ТипОтчетаСтрокой = ВариантыОтчетов.ТипОтчетаСтрокой(НастройкиОтчета.ОтчетСсылка);
    Если ТипОтчетаСтрокой = Неопределено Тогда
    Информация = ВариантыОтчетов.ИнформацияОбОтчете(НастройкиОтчета.ПолноеИмя, Истина);
    НастройкиОтчета.ОтчетСсылка = Информация.Отчет;
    КонецЕсли;
    КонецПроцедуры

    &НаСервере
    Процедура ПриЗагрузкеВариантаНаСервере(Настройки)

    ОбновитьИнформациюОВариантеОтчета();

    АвтоНавигационнаяСсылка = Ложь;
    НавигационнаяСсылка = ";
    Если ЗначениеЗаполнено(НастройкиОтчета.ВариантСсылка)
    Тогда
    НавигационнаяСсылка = ПолучитьНавигационнуюСсылку(НастройкиОтчета.ВариантСсылка);
    КонецЕсли;

    КонецПроцедуры

    &НаСервере
    Процедура ОбновитьИнформациюОВариантеОтчета()

    Запрос = Новый Запрос;
    Запрос.Текст =
    "ВЫБРАТЬ РАЗРЕШЕННЫЕ ПЕРВЫЕ 1
    | ВариантыОтчетов.Ссылка КАК ВариантСсылка,
    | ВариантыОтчетов.ПредопределенныйВариант.КлючЗамеров КАК КлючЗамеров,
    | ВариантыОтчетов.ПредопределенныйВариант КАК ПредопределенныйСсылка,
    | ВЫБОР
    | КОГДА ВариантыОтчетов.Пользовательский
    | ИЛИ ВариантыОтчетов.Родитель.КлючВарианта ЕСТЬ NULL
    | ТОГДА ВариантыОтчетов.КлючВарианта
    | ИНАЧЕ ВариантыОтчетов.Родитель.КлючВарианта
    | КОНЕЦ КАК ИмяИсходногоВарианта,
    | ВариантыОтчетов.Пользовательский КАК Пользовательский
    |ИЗ
    | Справочник.ВариантыОтчетов КАК ВариантыОтчетов
    |ГДЕ
    | ВариантыОтчетов.Отчет = &Отчет
    | И ВариантыОтчетов.КлючВарианта = &КлючВарианта";
    Запрос.УстановитьПараметр("Отчет", НастройкиОтчета.ОтчетСсылка);
    Запрос.УстановитьПараметр("КлючВарианта", КлючТекущегоВарианта);

    Выборка = Запрос.Выполнить().Выбрать();

    Если Выборка.Следующий() Тогда
    НастройкиОтчета.Вставить("ВариантСсылка", Выборка.ВариантСсылка);
    НастройкиОтчета.Вставить("КлючЗамеров", Выборка.КлючЗамеров);
    НастройкиОтчета.Вставить("ПредопределенныйСсылка", Выборка.ПредопределенныйСсылка);
    НастройкиОтчета.Вставить("ИмяИсходногоВарианта", ?(Выборка.Пользовательский, Выборка.ИмяИсходногоВарианта, КлючТекущегоВарианта));
    НастройкиОтчета.Вставить("Пользовательский", Выборка.Пользовательский);
    Иначе
    НастройкиОтчета.Вставить("ВариантСсылка", Неопределено);
    НастройкиОтчета.Вставить("КлючЗамеров", Неопределено);
    НастройкиОтчета.Вставить("ПредопределенныйСсылка", Неопределено);
    НастройкиОтчета.Вставить("ИмяИсходногоВарианта", Неопределено);
    НастройкиОтчета.Вставить("Пользовательский", Неопределено);
    КонецЕсли;
    КонецПроцедуры
```